
//***************************************************************
//* MACRO: WATERMRK.WCM
//* PURPOSE: Helps user create a watermark from a .wpg file or from text.
//***************************************************************
Application (A1; "WordPerfect"; Default; "EN")

If (?Substructure)
MessageBox (; "Not At Main Edit Screen"; "This macro can be played only from the 
main editing screen.  Exit all substructures (headers, text boxes, etc.) and play the macro 
again."; IconStop!)
Quit

EndIf

DefaultFontSize := 48p // Default font size
WtmkA := 1 // Default to use watermark A
WpgWtmk := 1 // Default to graphic watermark
Path := ?PathGraphics
If (Path = "")

Path := ?PathGraphicsSupplemental
EndIf
If (Path = "")

Path := ?PathProgram
Call (CheckDir@)

EndIf
If (SubStr (Path; StrLen (Path); 1) <>"\")

Path := Path + "\"
EndIf
HeightOfBox := ?PaperLength
WidthOfBox := ?PaperWidth

Label (MainDialog@)
DialogDefine ("Dlg1"; 50; 50; 154; 68; 1 + 2 + 16; "Watermark Options")
DialogAddGroupBox ("Dlg1"; "Gb2"; 8; 4; 58; 38; "T&ype")
DialogAddRadioButton ("Dlg1"; "Rb5"; 18; 14; 38; 11; "&Graphic"; WpgWtmk)
DialogAddRadioButton ("Dlg1"; "Rb4"; 18; 26; 38; 11; "&Text"; TxtWtmk)
DialogAddGroupBox ("Dlg1"; "Gb3"; 74; 4; 72; 38; "&Placement")
DialogAddRadioButton ("Dlg1"; "Rb6"; 84; 14; 53; 11; "Watermark &A"; WtmkA)
DialogAddRadioButton ("Dlg1"; "Rb7"; 84; 26; 53; 11; "Watermark &B"; WtmkB)
Display (On!)
DialogDisplay ("Dlg1"; "Rb5")
DialogResult := MacroDialogResult
DialogDestroy ("Dlg1")
Display (Off!)
If (DialogResult = 2)

Go (End@)
EndIf



If (TxtWtmk = 1)
Call (TextWatermark@)

Else
Call (GraphicWatermark@)

EndIf

// Do this only at the main document screen
While (?Substructure <>0)
SubstructureExit ()
EndWhile
MacroStatusPrompt (On!; "Placing watermark...")
If (WtmkA = 1)

If (?WatermarkAOccur <>0)
StructureName := "Watermark A"
Call (StructureExists@)

EndIf
WatermarkA (Create!)

Else
If (WtmkB = 1)

If (?WatermarkBOccur <>0)
StructureName := "Watermark B"
Call (StructureExists@)

EndIf
WatermarkB (Create!)

EndIf
EndIf

If (TxtWtmk = 1)
CenterCurrentPage (On!)
Justification (Center!)
FontSize (DefaultFontSize)
Type (WatermarkText)
Display (On!)
MessageBox (MsgBoxResult; "Watermark Text Options"; "The current font is ^0, and the
current size is ^1.  Would you like to change these options?"; IconQuestion! | YesNo! | 
HasParameters!; ?Font; DefaultFontSize)
Display (Off!)
If (MsgBoxResult = 6)

PosDocTop ()
FontDlg ()

EndIf
Else

BoxCreate (WatermarkImageBox!)
BoxAttachTo (Page!)
BoxHorizontalAlignment (AlignMargins!; FullAlign!; 0.0")



BoxVerticalAlignment (FullAlign!; 0.0")
BoxBorder (0)
BoxContentType (Image!)
OnError Call(ImageError@)
BoxImageRetrieve (MakeInternal!; Watermark)
BoxEnd (Save!)

EndIf
SubstructureExit ()

Label (End@)
Quit

//***************************************************************
//* ROUTINE: TextWatermark@
//* INPUT VARIABLES: none
//* OUTPUT VARIABLES: 
//* DESCRIPTION: Displays a dialog for entering watermark text.
//***************************************************************
Label (TextWatermark@)
DialogDefine ("Dlg1"; 50; 50; 166; 112; 1 + 2 + 16; "Watermark Creation Macro")
DialogAddEditBox ("Dlg1"; "Eb1"; 8; 8; 150; 50; 64 + 32; WatermarkText; 1024)
DialogAddHLine ("Dlg1"; "Hl1"; 6; 66; 154)
DialogAddText ("Dlg1"; "T1"; 8; 70; 150; 20; 1; "Enter text to place in the watermark.  Use Ctrl-
Enter to start a new line.")
Display (On!)
DialogDisplay ("Dlg1"; "Eb1")
DialogResult := MacroDialogResult
DialogDestroy ("Dlg1")
Display (Off!)
If (DialogResult = 2)

Go (End@)
EndIf
Return
//***************************************************************

//***************************************************************
//* ROUTINE: GraphicWatermark@
//* INPUT VARIABLES: none
//* OUTPUT VARIABLES: 
//* DESCRIPTION: Displays a dialog for selecting a watermark graphic.
//***************************************************************
Label (GraphicWatermark@)

Watermark=FileNameDialog (OpenDialog! | FileMustExist!; "Open Watermark"; ; Path; 
""; "*.*")
If(Watermark="")

Wait(1)



MessageBox (rVal;"Select Watermark";"You must select a valid graphic 
file.";Style:OKCancel!|IconExclamation! )
If(rVal=1) // OK

Go (GraphicWatermark@)
Else //Cancel

Go (MainDialog@)
EndIf

EndIf
Return
//***************************************************************

//***************************************************************
//* ROUTINE: StructureExists@
//* INPUT VARIABLES: StructureName
//* OUTPUT VARIABLES: 
//* DESCRIPTION: Displays a message box.
//***************************************************************
Label (StructureExists@)
MessageBox (MsgBoxResult; StructureName + " Already Exists"; "^0 already exists.  If you 
continue, the new watermark will replace the current one."; IconExclamation! | OKCancel! | 
HasParameters!; StructureName)
If (MsgBoxResult <>1)

Go (MainDialog@)
EndIf
Return
//***************************************************************

//***************************************************************
//* ROUTINE: CheckDir@
//* INPUT VARIABLES: Path
//* OUTPUT VARIABLES: Path
//* DESCRIPTION: This will create a dialog to prompt for a directory.
//***************************************************************
Label (CheckDir@)
DialogDefine ("PathDlg"; 50; 50; 165; 72; 1 + 2 + 16; "Graphic Directory")
DialogAddFilenameBox ("PathDlg"; "Fb1"; 8; 8; 150; 14; 1; Path; Path; "*.*")
DialogAddText ("PathDlg"; "T1"; 8; 28; 150; 20; 1; "Please select the directory that contains the 
Watermark graphics files.")
Label (Redisplay@)
Display (On!)
DialogDisplay ("PathDlg"; "Fb1")
Display (Off!)
If (MacroDialogResult = 2)

DialogDestroy ("PathDlg")
Go (End@)

EndIf



If (Path = "")
Go (Redisplay@)

EndIf
DialogDestroy ("PathDlg")
If (SubStr (Path; StrLen (Path); 1) <>"\")

Path := Path + "\"
EndIf
Return
//***************************************************************

//******** Error opening image *************
Label(ImageError@)

MessageBox(rVal;"Error";"There was an error opening "+Watermark+". Make sure it is a
valid graphic file and try again.";OK!|ICONEXCLAMATION!)
BoxEnd(Cancel!)
SubstructureExit()
If(WtmkA=1)

WatermarkA (Action: Off!)
Else

WatermarkB(Action: Off!)
EndIf
Quit

Return


